
ABSTRACT

Almost any report can be generated in at least two ways using
SAS software, and almost every report can be written using the
DATA step report writing capabilities (using the FILE-PUT mode),
but why reinvent the wheel if a PROC already exists that will give
you the desired report with a minimal amount of effort?  This
presentation will explore various common types of reports and
what factors you should consider in deciding which procedure is
best suited to your needs.  It is not meant to be a detailed guide
about PROC statement coding, but rather a guide on how to
decide which procedure fits the needs of a given situation.
Examples will be presented of differing types of reports with
discussion of the most efficient ways of programming those
reports in terms of programmer time.

INTRODUCTION

The versatility of SAS software allows almost identical reports to
be produced in multiple ways – the difference may be in how
many hours you want to spend writing the code.  Another factor
that may come into play is the CPU time used each time you
produce the report.  If you are running a batch job daily or weekly
on a very large data set, it is probably worthwhile to spend more
programming time to develop a more efficient program.  This
paper is not meant to show you CPU efficiency comparisons.
These are dependent on the size of your data sets, your
operating system, what kind of report you are trying to generate,
and a number of other factors.  In order to check CPU efficiency
you need to benchmark your programs.  This is much less critical
if you are working on a PC or on an ad-hoc or one-time report.

What issues do you need to consider when deciding what PROC
will produce your report with the least amount of programming
effort on your part?  The following are some of the most important
issues.

  • Is it a detail or summary report?
  • What information do you need in the report?
  • Do you need a specific layout or customized information?
  • Do you need cross-tabulations or hierarchical groupings?

Let’s consider each of these items to see which PROC(s) will let
you produce the desired report most easily.  All of the procedures
discussed in this paper are available in Base SAS.  Output from
the code given in this paper is shown in the Appendix.  (Note:
Tables 2 through 5 only include the last page of output generated
by these procedures.)

IS IT A DETAIL OR SUMMARY REPORT?

What is the difference between a detail report and a summary
report?  A detail report gives you at least one row of information
in your report for every observation in the data set (assuming you
are not using a WHERE statement in the procedure to limit which
observations are processed.)  A summary report generally does
some form of collapsing of observations so that each row in the
report represents information from more than one observation.

Which procedures produce what types of report?  In general:

  • PROC PRINT produces a detail report, although it can
     include summary information across observations
     overall and/or within subgroups.  PROC PRINT can
     also produce a summary report if the data has been
     outputted to a summary data set through a procedure
     like PROC MEANS, PROC SUMMARY or PROC
     UNIVARIATE.
  • PROC TABULATE produces summary tables.
  • PROC REPORT produces detail or summary reports.

This presentation focuses on the above procedures.

DOES ADDITIONAL INFORMATION OR DATA
NEED TO BE INCLUDED?

Do you need additional columns of information in the report that do
not already exist as variables in the data set?  An example of this
might be the sum or average of two existing variables in the data
set.  If that information is only necessary for the report, why go
through an additional DATA step to create a new variable?  PROC
REPORT will allow you to create new columns of information within
the PROC step.

If you need the same information represented in more than one way
in a report (for example, a medical code represented as a coded
value in the data set as well as an interpreted or formatted value),
then PROC REPORT will allow you to show both columns in the
same report.  PROC PRINT will also do this, but requires that you
create a new variable in the DATA step.  (It will not print the same
variable with two different formats in the same report.)  Using PROC
FORMAT to create a format to associate with the coded value in
conjunction with PROC REPORT will allow you to use both the
coded and formatted value in the same report without creating a
new variable.

The following code sets up the formats for selected variables, uses
PROC DATASETS to display the descriptor portion of the
permanent SAS data set SCPAPER.MBP, and creates the
temporary data set, WORK.MBP2.

LIBNAME SCPAPER 'C:\SAS\SCPAPER';

OPTIONS  LINESIZE=142 PAGESIZE=42 NODATE NONUMBER;

PROC FORMAT;
 VALUE $GENDER
       'M'='Male'
       'F'='Female';
 VALUE $RACE
       '1'='White'
       '2'='Black'
       '3'='Oriental'
       '4'='Other';
 VALUE $RCODE
   '001'='Baseline DBP > 90, decreased by 10% or more'
   '002'='Baseline DBP > 90, decreased by 5% up to 10% '
   '003'='Baseline DBP > 90, decreased by 0% up to 5% '
   '004'='Baseline DBP > 90, increased'
   '005'='Baseline DBP <= 90';



 VALUE NCENTER
       1='Dr. Green'
       2='Dr. Jones'
       3='Dr. Wright';

RUN;

**** CONTENTS OF ORIGINAL DATA SET ****;
**** SCPAPER.MBP (TABLE 1) ****;

PROC DATASETS LIB=SCPAPER;
 CONTENTS DATA=MBP;
TITLE1 'TABLE 1';
TITLE2 'CONTENTS OF SCPAPER.MBP DATA SET';
RUN;

**** CREATING WORK DATA SET OF MBP2 ****;

DATA MBP2;
 SET SCPAPER.MBP;
 CSTDIAM=STDIAM-BSTDIAM;
 LABEL CSTDIAM='Change in Diastolic BP';
RUN;

Appendix Table 1 shows the six demographic and diastolic blood
pressure variables present in the original data set,
SCPAPER.MBP.   The data set WORK.MBP2 contains these
variables, plus an additional created variable – CSTDIAM
(Change in Diastolic Blood Pressure).  Tables 2 and 3 present
detail reports using PROC PRINT and PROC REPORT,
respectively.  Table 3 includes two additional columns of
information generated with PROC REPORT.  The Physician
Name column is created based on the value for CENTER, and
Change in Diastolic Blood Pressure is computed with a
COMPUTE block.  The code that generated Table 3 also creates
a new data set called WORK.NEWMBP which includes the newly
computed variables.  The following code produces Appendix
Tables 2 and 3.

**** PROC PRINT FOR WORK.MBP2 ****;
**** DETAIL REPORT (TABLE 2) ****;

PROC PRINT DATA=MBP2 SPLIT='/' NOOBS;
 VAR PTNO CENTER NRACE GENDER
     BSTDIAM STDIAM CSTDIAM;
 FORMAT NRACE $RACE. GENDER $GENDER.;
 LABEL PTNO='Patient/Number'
       CENTER='Center'
       NRACE='Race'
       GENDER='Gender'
       BSTDIAM='Baseline/Diastolic/BP'
       STDIAM='Final/Diastolic/BP'
       CSTDIAM='Change in/Diastolic/BP';
 TITLE1 'TABLE 2';
 TITLE2
 'PROC PRINT OF WORK.MBP2 - DETAIL REPORT';
RUN;

**** PROC REPORT FOR SCPAPER.MBP ****;
**** DETAIL REPORT (TABLE 3) ****;
**** CREATES OUTPUT DATA SET WORK.NEWMBP ****;

PROC REPORT DATA=SCPAPER.MBP NOWD
        HEADSKIP SPACING=3 OUT=NEWMBP;
  COLUMN PTNO CENTER CENTER=NCENTER NRACE
        GENDER BSTDIAM STDIAM CSTDIAM;
   DEFINE PTNO/DISPLAY WIDTH=7 FORMAT=3.
        CENTER 'Patient Number';
   DEFINE CENTER/DISPLAY WIDTH=6 FORMAT=1.
        CENTER 'Center';
   DEFINE NCENTER/WIDTH=10 FORMAT=NCENTER.
        CENTER 'Physician Name';
   DEFINE NRACE/DISPLAY WIDTH=9 CENTER
        FORMAT=$RACE. 'Race';
   DEFINE GENDER/DISPLAY WIDTH=6 FORMAT=$GENDER.
       'Gender';
   DEFINE BSTDIAM/ANALYSIS MEAN WIDTH=9

       FORMAT=3. CENTER SPACING=5
       'Baseline/Diastolic/BP';
   DEFINE STDIAM/ANALYSIS MEAN WIDTH=9 FORMAT=3.
       CENTER 'Final/Diastolic/BP';
   DEFINE CSTDIAM/COMPUTED WIDTH=9 FORMAT=3.
       CENTER 'Change in/Diastolic/BP';
 COMPUTE CSTDIAM;
  CSTDIAM=STDIAM.MEAN-BSTDIAM.MEAN;
 ENDCOMP;
TITLE1 'TABLE 3';
TITLE2
 'PROC REPORT OF SCPAPER.MBP - DETAIL REPORT';
RUN;

DO YOU NEED A SPECIFIC LAYOUT OR
CUSTOMIZED INFORMATION?

This could include, but is not limited to the following:

  • providing customized information for subgroups
     or at the beginning or end of a report,
  • underlining column headers,
  • allowing variable values to extend across more than
     one line.

When developing a detail report, the simplest option you should
consider is PROC PRINT.  If you need sums for subgroups or at the
end of a report, PROC PRINT will give these to you.  However, if in
addition to or instead of sums, you need other statistics or
information (which can include customized text), then PROC
REPORT is probably the more appropriate approach.  Also, if you
need to underline the column headers or you have variable values
that will extend past the page width (based on the number of
variables you are trying to print and the PAGESIZE= option), then
PROC REPORT will provide these capabilities.

Appendix Tables 4 and 5 present detail tables, although Table 5
also includes summary information.  Table 4 uses PROC PRINT to
produce a detail report on WORK.MBP3 (which was created in a
data step from WORK.MBP2 so that the physician name and
comment variables were available for printing).  PROC REPORT
was used to generate Table 5.  In this table the column headers are
underlined, the comment variable values can extend across multiple
lines, and customized text with summary statistics are provided
overall and for each center.  The following code produces these
tables.

**** CREATING DATA SET WORK.MPB3 ****;
**** AND PROC PRINT ****;
**** DETAIL REPORT (TABLE 4) ****;

DATA MBP3; SET MBP2;
 LENGTH COMMENT $ 3;
 NCENTER=CENTER;
 RATIO=CSTDIAM/BSTDIAM;
 IF BSTDIAM GT 90 THEN DO;
  IF RATIO NE . AND RATIO LE -.10
     THEN COMMENT='001';
   ELSE IF -.10 LT RATIO LE -.05
     THEN COMMENT='002';
   ELSE IF -.05 LT RATIO LE 0
     THEN COMMENT='003';
  ELSE IF RATIO GT 0 THEN COMMENT='004';
 END;
 ELSE IF BSTDIAM LE 90 THEN COMMENT='005';
 FORMAT NCENTER NCENTER. COMMENT $RCODE.;
RUN;

PROC SORT DATA=MBP3;
 BY CENTER NCENTER;
RUN;

PROC PRINT DATA=MBP3 SPLIT='/' NOOBS;



 ID CENTER NCENTER;
 BY CENTER NCENTER;
 VAR PTNO NRACE GENDER BSTDIAM
     STDIAM CSTDIAM COMMENT;
 FORMAT NRACE $RACE. GENDER $GENDER.;
 LABEL PTNO='Patient/Number'
       CENTER='Center'
       NRACE='Race'
       GENDER='Gender'
       COMMENT='Comment'
       NCENTER='Physician/Name'
       BSTDIAM='Baseline/Diastolic/BP'
       STDIAM='Final/Diastolic/BP'
       CSTDIAM='Change in/Diastolic/BP';
 TITLE1 'TABLE 4';
 TITLE2
  'PROC PRINT OF WORK.MBP3 - DETAIL REPORT';
RUN;

**** PROC REPORT FOR SCPAPER.MBP ****;
**** WITH SUMMARY INFORMATION ****;
**** DETAIL REPORT (TABLE 5) ****;

PROC REPORT DATA=SCPAPER.MBP NOWD
            HEADSKIP HEADLINE SPACING=3;
 COLUMN CENTER CENTER=NCENTER PTNO NRACE GENDER
      BSTDIAM STDIAM CSTDIAM COMMENT;
  DEFINE PTNO/DISPLAY WIDTH=7 FORMAT=3.
       CENTER 'Patient Number';
  DEFINE CENTER/ORDER WIDTH=6 FORMAT=1.
       CENTER 'Center';
  DEFINE NCENTER/ORDER WIDTH=10 FORMAT=NCENTER.
       'Physician Name';
  DEFINE NRACE/DISPLAY WIDTH=9 CENTER
       FORMAT=$RACE. 'Race';
  DEFINE GENDER/DISPLAY WIDTH=6 FORMAT=$GENDER.
       'Gender';
  DEFINE BSTDIAM/ANALYSIS MEAN WIDTH=9 FORMAT=3.
       CENTER SPACING=5 'Baseline/Diastolic/BP';
  DEFINE STDIAM/ANALYSIS MEAN WIDTH=9 FORMAT=3.
       CENTER 'Final/Diastolic/BP';
  DEFINE CSTDIAM/COMPUTED WIDTH=9 FORMAT=3.
       CENTER 'Change in/Diastolic/BP';
  DEFINE COMMENT/COMPUTED WIDTH=25 SPACING=5
       FLOW FORMAT=$RCODE. 'Comment';

 COMPUTE CSTDIAM;
  CSTDIAM=STDIAM.MEAN-BSTDIAM.MEAN;
 ENDCOMP;

 COMPUTE COMMENT/CHARACTER;
  IF BSTDIAM.MEAN GT 90 AND
    (CSTDIAM/BSTDIAM.MEAN) LE -.10
       THEN COMMENT='001';
  ELSE IF BSTDIAM.MEAN GT 90 AND
    -.10 LT (CSTDIAM/BSTDIAM.MEAN) LE -.05
       THEN COMMENT='002';
  ELSE IF BSTDIAM.MEAN GT 90 AND
    -.05 LT (CSTDIAM/BSTDIAM.MEAN) LE 0
       THEN COMMENT='003';
  ELSE IF BSTDIAM.MEAN GT 90 AND
    (CSTDIAM/BSTDIAM.MEAN) GT 0
       THEN COMMENT='004';
  ELSE IF BSTDIAM.MEAN LE 90 THEN COMMENT='005';
 ENDCOMP;

 BREAK AFTER CENTER/SKIP;
 COMPUTE AFTER CENTER;
  PCHANGE=(CSTDIAM/BSTDIAM.MEAN)*100;
  LINE @3 ' ';
  LINE @22 111*'-';
  LINE @22
    '|The average change in diastolic blood'
    @61 'pressure for'
    @74 CENTER NCENTER. @85 'was'
    @89 CSTDIAM 6.2
    @96 'with an average % change of'
    @124 PCHANGE 7.2 @132 '|';
  LINE @22 111*'-';
  LINE @3 ' ';

 ENDCOMP;

 RBREAK AFTER/SKIP;
 COMPUTE AFTER;
  PCHANGE=(CSTDIAM/BSTDIAM.MEAN)*100;
  LINE @3 ' ';
  LINE @22 103*'=';
  LINE @22
    '|The average change in diastolic blood'
    @61 'pressure overall'
    @78 'was' @82 CSTDIAM 6.2
    @89 'with an average % change of'
    @116 PCHANGE 7.2 @124 '|';
 LINE @22 103*'=';
 LINE @3 ' ';
ENDCOMP;

TITLE1 'TABLE 5';
TITLE2
   'PROC REPORT OF SCPAPER.MBP - DETAIL REPORT';
RUN;

DO YOU NEED CROSS-TABULATIONS OR
HIERARCHICAL GROUPINGS?

Both of these types of tables are summary reports where you
display some sort of statistic(s) within a combination of variable
values in either the row and/or column dimensions.  PROC PRINT
can produce this type of report after outputting the summary
information from some other procedure.  However, this type of
report is usually generated with PROC TABULATE or PROC
REPORT because of the greater flexibility and need for fewer steps.

Appendix Tables 6, 7 and 8 all show the diastolic blood pressure
summarized by demographic data.  Table 6 provides the same
information as the other tables, but may not be as desirable for
reporting purposes.  Tables 7 and 8 produce very similar reports.
Table 7 (PROC TABULATE) requires less programming code, while
Table 8 (PROC REPORT) allows more flexibility in spacing between
columns of information, underlining of headers, and customized text
(if desired).  The following code generates Appendix Tables 6, 7 and
8.

**** PROC MEANS-PRINT FOR WORK.NEWMBP2 ****;
**** SUMMARY REPORT (TABLE 6) ****;

PROC MEANS DATA=MBP2 MAXDEC=2 NOPRINT;
 CLASS GENDER NRACE;
 VAR BSTDIAM STDIAM CSTDIAM;
 OUTPUT OUT=NEWMBP2 N=BSTDIAM STDIAM CSTDIAM
                   MEAN=MEANBST MEANST CMEANST
                   STD=STDBST STDST CSTDST;
RUN;

PROC PRINT DATA=NEWMBP2 LABEL NOOBS;
 VAR GENDER NRACE _TYPE_ BSTDIAM MEANBST STDBST
     STDIAM MEANST STDST CSTDIAM CMEANST CSTDST;
 FORMAT MEANBST MEANST CMEANST
     STDBST STDST CSTDST 8.2;
 LABEL BSTDIAM='Baseline Diastolic BP - N'
     STDIAM='Final Diastolic BP - N'
     CSTDIAM='Change in Diastolic BP - N'
     MEANBST='Baseline Diastolic BP - Mean'
     MEANST='Final Diastolic BP - Mean'
     CMEANST='Change in Diastolic BP-Mean'
     STDBST='Baseline Diastolic BP - Std. Dev.'
     STDST='Final Diastolic BP - Std. Dev.'
     CSTDST='Change in Diastolic BP - Std. Dev.';
 TITLE1 'TABLE 6';
 TITLE2
    'PROC MEANS-PRINT OUTPUT OF WORK.NEWMBP2';
 TITLE3 'SUMMARY REPORT';
RUN;



**** PROC TABULATE FOR WORK.MBP2 ****;
**** SUMMARY REPORT (TABLE 7) ****;

OPTIONS MISSING='0';

PROC TABULATE DATA=MBP2 NOSEPS;
 CLASS GENDER NRACE;
 VAR BSTDIAM STDIAM CSTDIAM;
 TABLE (NRACE ALL),GENDER*
      (BSTDIAM*(N*F=3. MEAN*F=6.2 STD*F=6.2)
      STDIAM*(N*F=3. MEAN*F=6.2 STD*F=6.2)
      CSTDIAM*(N*F=3. MEAN*F=6.2 STD*F=6.2))
      /RTS=15;
 KEYLABEL ALL='All'
          N='No.'
          MEAN='Mean'
          STD='Std Dev.';
 LABEL BSTDIAM='Baseline Diastolic BP'
       STDIAM='Final Diastolic BP'
       CSTDIAM='Change in Diastolic BP';
 FORMAT NRACE $RACE. GENDER $GENDER.;
 TITLE1 'TABLE 7';
 TITLE2
     'PROC TABULATE OUTPUT OF WORK.MBP2';
 TITLE3 'SUMMARY REPORT';
RUN;

**** PROC REPORT FOR WORK.NEWMBP ****;
**** SUMMARY REPORT (TABLE 8) ****;

PROC REPORT DATA=NEWMBP NOWD HEADSKIP
       HEADLINE SPACING=1;
 COLUMN NRACE (GENDER,
   (( 'Baseline/Diastolic BP'
   BSTDIAM BSTDIAM=MEANBST BSTDIAM=STDBST)
   ( 'Final/Diastolic BP'
   STDIAM STDIAM=MEANST STDIAM=STDST)
   ( 'Change in/Diastolic BP'
   CSTDIAM CSTDIAM=CMEANST CSTDIAM=CSTDST)));
  DEFINE NRACE/GROUP ORDER=INTERNAL WIDTH=6
       FORMAT=$RACE.;
  DEFINE GENDER/ACROSS FORMAT=$GENDER.
       SPACING=2 '_Gender_';
  DEFINE BSTDIAM/ANALYSIS N WIDTH=3 FORMAT=3.
       CENTER SPACING=3 'No.';
  DEFINE MEANBST/ANALYSIS MEAN WIDTH=8
       FORMAT=8.2 CENTER 'Mean';
  DEFINE STDBST/ANALYSIS STD WIDTH=6
       FORMAT=6.2 CENTER 'Std. Dev.';
  DEFINE STDIAM/N WIDTH=3 FORMAT=3.
       CENTER SPACING=3 'No.';
  DEFINE MEANST/MEAN WIDTH=8 FORMAT=8.2
       CENTER 'Mean';
  DEFINE STDST/ STD WIDTH=6 FORMAT=6.2
       CENTER 'Std. Dev.';
  DEFINE CSTDIAM/ANALYSIS N WIDTH=3 FORMAT=3.
       CENTER SPACING=3 'No.';
  DEFINE CMEANST/ANALYSIS MEAN WIDTH=8
       FORMAT=8.2 CENTER 'Mean';
  DEFINE CSTDST/ANALYSIS STD WIDTH=6
       FORMAT=6.2 CENTER 'Std. Dev.';

 BREAK AFTER NRACE/SKIP;

 RBREAK AFTER/ SKIP SUMMARIZE;

 COMPUTE AFTER;
  NRACE='All';
 ENDCOMP;
 TITLE1 'TABLE 8';
 TITLE2
    'PROC REPORT OUTPUT OF WORK.NEWMBP';
 TITLE3 'SUMMARY REPORT';
RUN;

CONCLUSION

This paper has attempted to demonstrate some of the versatility
available in SAS when generating reports.  PROC PRINT, PROC
TABULATE and PROC REPORT are just several of the many ways
to produce a report using SAS software.  An additional procedure in
Base SAS that can be used to generate reports is PROC SQL.  This
procedure was not covered in this paper due to time and space
limitations, but should also be considered for report-writing.
Hopefully, this paper will help guide you on which approach will give
you the report you desire with the least amount of programming
effort.

GHFUHDVHG E\ �� XS WR ��

�� %ODFN 0DOH �� �� �� %DVHOLQH '%3 ! ���
GHFUHDVHG E\ �� XS WR ��

�� :KLWH 0DOH �� �� �� %DVHOLQH '%3 ! ���
GHFUHDVHG E\ �� XS WR ��

�� :KLWH )HPDOH �� �� �� %DVHOLQH '%3 ! ���
GHFUHDVHG E\ �� XS WR ���

�� :KLWH )HPDOH ��� �� �� %DVHOLQH '%3 ! ���
GHFUHDVHG E\ �� XS WR ��

�� :KLWH )HPDOH �� �� �� %DVHOLQH '%3 ! ���
GHFUHDVHG E\ �� XS WR ��

�� :KLWH )HPDOH �� �� � %DVHOLQH '%3 ! ���
LQFUHDVHG

�� :KLWH )HPDOH �� �� �� %DVHOLQH '%3 ! ���
GHFUHDVHG E\ �� XS WR ���

�� :KLWH )HPDOH �� �� � %DVHOLQH '%3 ! ���
LQFUHDVHG

���������������������������������������������������������������������������������������������������������������
_7KH DYHUDJH FKDQJH LQ GLDVWROLF EORRG SUHVVXUH IRU 'U� :ULJKW ZDV ����� ZLWK DQ DYHUDJH � FKDQJH RI ����� _
���������������������������������������������������������������������������������������������������������������

                                                                                                       
_7KH DYHUDJH FKDQJH LQ GLDVWROLF EORRG SUHVVXUH RYHUDOO ZDV ����� ZLWK DQ DYHUDJH � FKDQJH RI ����� _

)HPDOH 2WKHU � � ������ ���� � ����� ���� � ����� ����
0DOH :KLWH � �� ����� ���� �� ����� ���� �� ����� ����
0DOH %ODFN � � ����� ���� � ����� ���� � ����� ����
0DOH 2WKHU � � ����� � � ����� � � ����� �

